NASA Technical Paper 1838

A Microprocessor-Based Table Lookup Approach for Magnetic Bearing Linearization

Nelson J. Groom and James B. Miller

MAY 1981

NASA

NASA Technical Paper 1838

A Microprocessor-Based Table Lookup Approach for Magnetic Bearing Linearization

Nelson J. Groom and James B. Miller
Langley Research Center
Hampton, Virginia

NASA
National Aeronautics
and Space Administration

Scientific and Technical
Information Branch

1981

SUMMARY

An approach for producing a linear transfer characteristic between force command and force output of a magnetic bearing actuator without flux biasing is presented. The approach is microprocessor based and uses a table lookup to generate drive signals for the magnetic bearing power driver. An experimental test setup used to demonstrate the feasibility of the approach is described, and test results are presented. The test setup contains bearing elements similar to those used in a laboratory model annular momentum control device (AMCD).

INTRODUCTION

This paper describes an approach for producing a linear transfer characteristic between the force command and force output of a magnetic bearing actuator. The approach, which is microprocessor based and uses a table lookup to generate drive signals for the magnetic bearing actuator power driver, was investigated for application to a laboratory model annular momentum control device (AMCD). The laboratory model (described in ref. 1) was built to investigate potential problem areas in implementing the AMCD concept and is being used as part of an AMCD hardware technology development program. The basic AMCD concept is that of a rotating annular rim, suspended by a minimum of three magnetic bearing suspension stations and driven by a noncontacting electromagnetic spin motor. A detailed discussion of the rationale for the AMCD configuration and of some of its potential applications is presented in reference 2.

As described in reference 1, the magnetic-bearing linearization technique used in the original laboratory model AMCD magnetic suspension was to differentially control sets of magnetic bearing elements about a permanent-magnet bias flux. Preliminary tests indicated that this approach (permanent-magnet flux biasing) presented a problem from a control system standpoint (ref. 3). For a given equivalent permanent-magnet stiffness, a minimum bearing servo bandwidth is required for stability. The existence of structural modes in the area of the laboratory model bearing servo crossover restricted the amount of bearing servo damping that could be achieved.

Because of the limitations of permanent-magnet flux biasing encountered with the laboratory model AMCD, a decision was made to explore an alternate approach to the design of the magnetic suspension system (ref. 4). As a result, a new magnetic suspension system for the laboratory model has been designed, fabricated, and tested (ref. 5). The new system uses a zero bias flux approach for the magnetic bearing actuators. Analog multiplier and square root modules produce a direct solution to the ideal magnetic actuator force equation to provide a linear force-current characteristic. (For further discussion of magnetic bearing actuator control approaches, see ref. 6.) The accuracy of this approach is limited by the accuracy with which the ideal force equation approximates the actual characteristics of the actuator and by the accuracy of the analog components. This paper presents a zero bias flux linearization approach which is digital and which uses a lookup table constructed from measured actuator characteristics. An experimental test setup that was used to develop this approach is described, and test results are presented.

SYMBOLS

Dimensional quantities are presented in both SI Units and U.S. Customary Units. Measurements were made in U.S. Customary Units.

F_B force produced by bottom electromagnet

F_C force command for magnetic bearing actuator

$F_C(m)$ value of force command associated with mth line segment

F_T force produced by top electromagnet

f_s microcomputer system sample rate

ΔF_C $= F_C(m+1) - F_C(m)$ for all m, where $m < N$

δF_C $= F_C - F_C(m)$, where $F_C(m) \leqq F_C < F_C(m+1)$

G displacement of suspended element with respect to centered position in magnetic bearing actuator gaps

G_B gap of bottom electromagnet

G_O magnetic bearing actuator gap with suspended element centered

G_T gap of top element

I_B current in bottom electromagnet

I_C current command for magnetic bearing actuator

$I_C(m)$ stored value of current command associated with mth line segment

I_T current in top electromagnet

K electromagnet constant

N number of line segments

SLOPE(m) slope of mth line segment from $(F_C(m), I_C(m))$ to $(F_C(m+1), I_C(m+1))$

Abbreviations:

A/D analog-to-digital converter

AMCD annular momentum control device

D/A digital-to-analog converter

dc direct current

emf electromotive force

APPROACH

Magnetic Bearing Control Approach

The magnetic bearing control approach is one which uses zero bias flux. Figure 1, a schematic representation of a magnetic bearing element pair, is



Figure 1.- Magnetic bearing actuator.

presented in order to describe this approach. Included in the figure are top and bottom electromagnets, with currents I_T and I_B, respectively; a portion of the suspended element which is centered between the electromagnets; and a position sensor which measures the displacement G of the suspended element with respect to the centered position G_o. In the zero bias flux control approach, one electromagnet at a time is controlled; the top electromagnet is controlled for an upward force and the bottom electromagnet is controlled for a downward force (since each electromagnet can physically produce only a unidirectional force). Figure 1 indicates that if up is taken as the positive direction, the electromagnet gaps are

$$G_T = G_O - G \tag{1}$$

and

$$G_B = G_O + G \tag{2}$$

Assuming negligible fringing and ignoring nonlinear core effects, the force produced by each electromagnet is given by (ref. 4)

$$F_T = K\left(\frac{I_T^2}{G_T^2}\right) \tag{3}$$

and

$$F_B = K\left(\frac{I_B^2}{G_B^2}\right) \tag{4}$$

The composite force-current characteristic of a zero bias flux actuator with the suspended element centered is shown in figure 2.



Figure 2.- Composite force-current characteristic of a zero bias flux magnetic actuator.

Linearization Approach

The linearization approach is microprocessor based and uses a lookup table. The most straightforward approach for producing a linear transfer characteristic between force command and force output would be to solve equations (3) and (4) for the electromagnet current required to produce the desired force. That is,

$$I_T = G_T\left(\frac{|F_C|}{K}\right)^{1/2} \qquad (F_C > 0) \qquad (5)$$

and

$$I_B = G_B\left(\frac{|F_C|}{K}\right)^{1/2} \qquad (F_C < 0) \qquad (6)$$

where F_C is the commanded force. Implementing the solutions of equations (5) and (6) digitally could produce a more computationally accurate result than the analog solution but would not account for deviation of the actual bearings from the ideal model. This solution would also require that considerable computing power be dedicated to a relatively minor portion of the total control system.

The table lookup approach, which employs a one-to-one correspondence between force command F_C and output current command I_C permits the output/input function to conform, within the limits of quantization error, to the actual bearing characteristics. However, this form of table requires considerable memory space. The minimum memory requirement is obtained by selecting the minimum set of straight line segments which approximate the output/input relationship within the desired tolerance. The major disadvantage of this choice of line segments is that considerable time may be spent searching for the appropriate line segment.

Since in this application computational time is more critical than memory size, an approximation was selected which requires more memory but eliminates the search time. The minimum set of equally spaced line segments N are selected so that $N = 2^n$, and the $N + 1$ points defining the line segments are separated by equal force command steps ΔF_C. If the range of F_C is scaled to a B bit binary word ($B > n$), the n most significant bits of F_C uniquely identify the N line segments. In actual computation these bits identify the lookup table data associated with the mth force command data point $F_c(m)$ such that $F_c(m) \leqq F_c < F_c(m+1)$. The $B - n$ least significant bits of F_c represent the difference δF_c between F_c and $F_c(m)$. (See fig. 3.)



Figure 3.- Table lookup approach.

The data contained in the lookup table consist of the slope SLOPE(m) of each line segment m between $(F_c(m),I_c(m))$ and $(F_c(m+1),I_c(m+1))$ and of the current command $I_c(m)$ corresponding to $F_c(m)$. The location of the appropriate data is obtained directly from the force command word by masking and shifting. Calculation of output current command I_c at $G = 0$ requires a single multiplication of δF_c by SLOPE(m) and addition of this result to $I_c(m)$. Since actual variation of electromagnet current with gap was observed to be linear, as indicated by equations (5) and (6), calculation of I_c for other values of G is accomplished by multiplying I_c by the appropriate bearing gap. For a more detailed description of the table lookup algorithm, see appendix A.

HARDWARE TESTS

Description of Test System

The test system consisted of a magnetic bearing test fixture connected to a microcomputer system as shown in figure 4. This system was used to obtain the data required to develop the lookup table and to obtain data on the performance of the proposed approach. A description of the current driver shown in figure 4 is given in appendix B.



Figure 4.- Magnetic bearing test system.

Magnetic bearing test fixture.- The magnetic bearing test fixture shown in figure 5 consists of a magnetic bearing element pair; an equivalent "suspended" element, which is connected through a pair of load cells to the base of the fixture; and a pair of position sensors. The suspended element can be set to any desired vertical position in the magnetic bearing gap using the adjusting screws mounted on the load cells. The position sensors are used to measure the bearing gaps. The magnetic bearing elements have the same dimensions as the original magnetic bearing elements delivered with the laboratory model AMCD which is described in reference 1. Two main differences exist between the test fixture bearing elements and the original laboratory model elements: (1) the core material of the test fixture bearing elements is SAE 1010 soft steel (as opposed to a lower loss silicon core iron used in the original elements), and (2) the test fixture bearing elements contain no permanent-magnet material.



L-80-3798

Figure 5.- Magnetic bearing test fixture.

The load cells, shown in figure 5, are strain-gage bridge instrumented bending beams. The output of the bridge is a voltage which is directly proportional to the load applied to the beam. The cells were connected as shown in figure 6. They have a load range of ±44.48 N (±10 lb) and a nominal scale factor of ±0.045 mV/N-V (±0.2 mV/lb-V). Scale factor and offset differ from cell to cell and vary with changes in test fixture configuration, power supply voltage, and temperature; therefore, software was designed to provide for periodic system calibration.

Calibration was accomplished by applying a sequence of known loads to each cell and performing a first-order least-squares fit to the resulting data. Typical raw calibration data for one cell are given in figure 7.



Figure 6.- Load cell connection diagram.



Figure 7.- Typical calibration data for a given load cell.

Microcomputer system.- The same microcomputer system was used for both characterization and performance tests. The major system components include (1) microcomputer and memory, (2) analog interface unit, (3) cartridge data recorder, and (4) portable data terminal (fig. 4).

The microcomputer is an Intel[1] 8080 microprocessor-based computer which includes serial and parallel ports, programmable hardware timers, and 2000 bytes of random access memory (RAM). This processor was selected for the initial phase of development because it is well supported in hardware and software. Available software includes FORTRAN as well as assembly language and the manufacturer's high-level language.

Most of the test programs were written in FORTRAN with assembly language hardware drivers. However, the table lookup routine was written in assembly language because of its time-critical nature. A listing of the table lookup routine is presented in appendix A. Memory expansion boards containing 24 000 bytes of RAM memory were added to permit run time storage of large FORTRAN programs and substantial test data.

The analog interface unit provides up to 32 multiplexed A/D input channels and two D/A output channels. The input channels provide the processor with position and force data from the test fixture position sensors and load cells and provide external command signals during those tests in which an analog force command is used. The two output channels are used to supply the current command to the bearing current driver. The analog interface unit was programmed to perform conversions in two modes. During characterization tests, conversions were initiated under program control, and a flag on the interface was set at end of conversion. During those portions of the performance test when the table lookup algorithm was running in a real-time environment, the interface was programmed to start a conversion sequence on the rising edge of a hardware timer and to interrupt the processor at the end of conversion. This technique provides a stable sampling rate and does not require software timing loops.

The data recorder was used for temporary storage and for transport of test programs and test results. This recorder permitted flexible use of the microcomputer system at a location which was remote from the microprocessor development system used for software generation.

The data terminal provided run time parameter selection, control, and monitoring of the system test programs.

Description of Tests

Characterization tests.- Characterization tests were required to establish the values of lookup table data. These data were collected by the microprocessor system and transferred to a programmable desk-top calculator for analysis

---

[1]Use of names of manufacturers in this report does not constitute an official endorsement of such manufacturers, either expressed or implied, by the National Aeronautics and Space Administration.

and reduction. A FORTRAN test program applied a 2048-point sequence of equally spaced current drive signals to the magnetic bearing test fixture and measured the actual force registered by the test fixture load cells at a manually selected rim position. This process was repeated for a series of rim positions. These positions were varied from -0.127 cm (-0.050 in.) to 0.127 cm (0.050 in.) in increments of 0.0127 cm (0.005 in.). The resulting force vs current data were converted by linear interpolation to a 2048-point current vs force table with equal force steps. These data were systematically reduced to form an N-point minimum size lookup table ($N = 2^n$) capable of approximating the measured data to within ±1 percent. This tolerance is within the limits of the existing analog solution to the ideal force equation.

Performance tests.- Two tests were performed to evaluate the assembly language interpolation routine and the table lookup technique:

1. Linearity tests (static) - A sequence of force commands were applied to the test fixture through the linearization algorithm by a FORTRAN test program. The sequence of force commands was applied in the same direction and over the same range as in the characterization tests. The force produced on the rim was measured by the load cells and compared with the input command.

2. Frequency response tests (dynamic) - A sinusoidal force-command signal was applied to the test fixture through the linearization algorithm, and the output current response was observed. The response measurements were performed by a frequency response analyzer. The force-command signal was given a dc offset so that measurements could be performed independently on upper and lower bearing elements. For this portion of the test, the characterization data were replaced by a linear table to permit direct comparison of input force command with current output. This replacement was necessitated by a mechanical resonance of the test fixture which prevented use of load cells for frequency response measurements. The frequency was measured over a bandwidth slightly larger than half the sampling frequency of the algorithm.

TEST RESULTS AND DISCUSSION

Characterization Tests

The force, current, and gap relationships obtained from the data collected during characterization tests are similar to those obtained from the ideal electromagnet force equations (eqs. (3) and (4)). However, measured data deviated from ideal relationships near the zero force points, and differences between constants for upper and lower bearing elements were observed. Figure 8 is a plot of the actual force data resulting from the application of a sequence of equally spaced current drive signals. Each curve represents 2048 data points



Figure 8.- Force-current data taken from magnetic bearing test fixture.

taken at a given gap setting. The force vs current data with equal current steps were converted by linear interpolation to current vs force data with equal force steps. Analysis of the resulting data indicated that the relationship between current and gap for a given force was sufficiently linear to permit gap compensation to be accomplished by multiplying the output current by the gap (as in eqs. (5) and (6)). Systematic reduction of the data for $G = 0$ resulted in a 128-point table that was capable of reproducing the original data to within 0.5 percent, as illustrated in figure 9. System memory requirements for this size table are quite reasonable.

Performance Tests

Linearity.- The main results of the table lookup algorithm linearity tests are shown in figures 10 and 11. Figure 10 shows the relation between force command input and the measured force output of the system. This particular plot is for $G = 0$, but the same result was obtained for other values of G. The actual percentage force error can be seen in figure 11. These results, although still within acceptable limits (less than ±1 percent error), do not agree completely with the errors predicted during data reduction (shown in fig. 9).



Figure 9.- Error associated with reduction of original data (for G = 0) to a 128-point table.



Figure 10.- Static output/input characteristics of table lookup algorithm.



Figure 11.- Percentage force error of table lookup algorithm.

In particular, relatively large errors occur at the maximum positive and negative excursions of the force command. Since no attempt was made to minimize the effects of hysteresis in the test fixture, considerable residual magnetic flux can exist in the electromagnet cores and in the equivalent "suspended" element. Variations in the magnetic history of the test fixture are the probable cause of these errors. One possible way to reduce the effects of hysteresis on accuracy would be to use flux feedback in the power driver loop instead of current feedback. Another more obvious approach would be to use low-hysteresis material in both the electromagnet core and the suspended element magnetic circuit material.

Frequency response.- The table lookup algorithm required approximately a 10-percent greater execution time than the 1.85-ms worst-case value predicted by evaluating the microprocessor instruction execution times. This loss of time occurred during the test because the table lookup algorithm instructions and data were stored in RAM memory on a memory expansion board. The increased execution time limited the system sample rate f_s to a maximum of approximately 490 Hz rather than the 540 Hz which was predicted.

Results of the frequency response test are summarized in figure 12. The bandwidth exceeds 100 Hz, which was considered adequate for this portion of the bearing system. The magnitude is flat from dc to approximately $f_s/4$ and then rolls off gradually to approximately $f_s/2$, which is the theoretical frequency limit for a sampled data system. Since this algorithm retains no history of the signals and has an almost fixed execution time, the phase response varies linearly with frequency to approximately $f_s/2$.



(a) Magnitude.



(b) Phase.

Figure 12.- Frequency response (magnitude and phase) of table lookup algorithm at 490-Hz repetition rate.

CONCLUDING REMARKS

A microprocessor based table lookup approach for magnetic bearing linearization without flux biasing has been presented, and an experimental test setup used to demonstrate the feasibility of the concept has been described. Results obtained with the experimental test setup generally showed very close agreement with theoretical predictions. Using a 128-point table, the table lookup algorithm produced a linear transfer characteristic between force command and force output of the test fixture magnetic bearing actuator to within ±1-percent error. The frequency response of the algorithm was greater than 100 Hz, which should be adequate for this portion of the bearing system.

This approach when used as an inner loop for the actuators could form the basis for an all-digital magnetic suspension control system. One application would be the laboratory model annular momentum control device (AMCD). An all-digital system would allow control-system parameter changes to be made in software without requiring the circuit component changes and circuit rewiring which are necessary with existing analog systems. Also, advanced controller design approaches could be more easily implemented.

Langley Research Center
National Aeronautics and Space Administration
Hampton, VA 23665
March 19, 1981

APPENDIX A

TABLE LOOKUP ALGORITHM

Algorithm Description

This appendix presents a listing, in assembly language, of the table look-up algorithm used to drive the magnetic bearing test fixture. The algorithm performs a nonlinear transformation of an input force command F_C and a measured rim displacement G into current commands I_C to either the top or bottom bearing element. The transformation approximates the solutions of equations (5) and (6) in the main text. The lookup table contains 128 four-byte data groups. Each of the groups consists of a 16-bit value of SLOPE(m) and a 16-bit value of current $I_C(m)$ obtained from measurement of the actual bearing force-current characteristic with the rim centered (i.e., $G_T = G_B = G_O$).

The routine is interrupt driven and is called upon completion of the force command A/D conversion which is initiated by a hardware real-time clock. After reading the force command, the processor sets up the analog interface and initiates conversion of the rim displacement. Because of the configuration of the analog interface, the force command F_C and rim displacement G are 12-bit two's complement numbers which are sign extended to 16 bits.

The least significant 5 bits of F_C are removed and stored, since they contain the value δF_C. The remaining 7 bits identify the 128 data groups. These bits are appropriately shifted, converted to offset binary, and added to the lookup-table base address to produce a memory pointer. The stored slope value is loaded using this memory pointer, and the pointer is incremented. The slope value is then multiplied by δF_C (an 8-bit by 16-bit multiplication routine is used to conserve time). This result is added to the stored value of $I_C(m)$, which is now addressed by the memory pointer. The result, $I_C(F_C,G_O)$, is the current command required to produce a desired force by either bearing element when the rim is centered ($G_T = G_B = G_O$). Since the required current is directly proportional to the bearing gap, the current command for any rim displacement is given by the following relationships:

$$I_T = |I_C|\left(\frac{G_T}{G_O}\right) \qquad (I_C \geq 0)$$

$$I_B = |I_C|\left(\frac{G_B}{G_O}\right) \qquad (I_C < 0)$$

where $I_C = f(F_C,G_O)$. After substitution of the table lookup algorithm for I_C and the expressions for G_T and G_B from equations (1) and (2) in the main text, these equations become

APPENDIX A

$$I_T = \left(I_C(m) + \text{SLOPE}(m)\ \delta F_C\right)\left(\frac{G_O - G}{G_O}\right)$$

$$I_B = \left(I_C(m) + \text{SLOPE}(m)\ \delta F_C\right)\left(\frac{G_O + G}{G_O}\right)$$

Since G_O is a constant, these equations can be rewritten as

$$I_T = \left[\left(\frac{I_C(m)}{G_O}\right) + \left(\frac{\text{SLOPE}(m)}{G_O}\right)\delta F_C\right](G_O - G)$$

$$I_B = \left[\left(\frac{I_C(m)}{G_O}\right) + \left(\frac{\text{SLOPE}(m)}{G_O}\right)\delta F_C\right](G_O + G)$$

The lookup table slope and current entries are predivided by G_O to eliminate a real-time multiplication by $1/G_O$.

Because of the scaling of inputs and outputs (see table A1) and the need to perform all calculations on 16-bit or less integer numbers, the stored look-up table values are scaled as shown in table A2.

TABLE A1.- SCALING OF INPUTS AND OUTPUTS

Variable	Range	Analog scale factor	Analog range	Digital scale factor	Digital range	Total scale factor
F_C	±11.12 N ±2.5 lb	0.9 V/N 4 V/lb	±10 V	204.8	±2047	184.2 819.2
I_C	±5 A	1 V/A	±5 V	409.6	±2047	409.6
G	±0.127 cm ±0.05 in.	[a]78.7 V/cm 200 V/in.	±10 V	[b]409.6	±4096	32.2×10^3 5×2^{14}

[a]These values include the sum of both position sensors.
[b]This value includes a software multiplication by 2 (line 95).

TABLE A2.- SCALING OF TABLE VALUE

Variable	Units	Scaling factors[a,b,c]
SLOPE(m)	A/N	$\frac{(0.5)(2^8)(2^{16})}{(G_o)(32.2)(10^3)}$
	A/lb	$\frac{(0.5)(2^8)(2^{16})}{(G_o)(5)(2^{14})}$
I_c(m)	A	$\frac{(2^{16})}{(G_o)(32.2)(2^{14})}$
	A	$\frac{(2^{16})}{(G_o)(5)(2^{14})}$

[a]The (0.5) factor accounts for the difference between the total scale factors for F_c and I_c.

[b]The (2^8) and (2^{16}) factors are employed to maximize the number of significant bits while maintaining intermediate and final results within a 16-bit integer format. These factors are removed by implied shifts during the two multiplications.

[c]Factors $(G_o)(32.2)(10^3)$ and $(G_o)(5)(2^{14})$ are for G_o in SI (cm) and U.S. Customary Units, respectively.

APPENDIX A

Program Listing

```
ASM80 :F1:LOOKUP.ASM PRINT(:F1:LOOKUP.LST) OBJECT(:F1:LOOKUP.OBJ)PAGEWIDTH(80)
                    EJECT


ISIS-II 8080/8085 MACRO ASSEMBLER, V3.0        MODULE   PAGE    1


 LOC  OBJ         LINE        SOURCE STATEMENT

                       ;
                       ;          LOOKUP. ASM
                       ;
                       ;
                              EXTRN CURPOS,CURNEG,ADCIN,TABLE,CONV,MUXADR,STAT
                         US

                       ;     SAVE MAIN PROGRAM REGISTERS
 0000 F5                 INTRU:  PUSH PSW        ;SAVE A & PSW
 0001 C5                         PUSH B          ;SAVE BC
 0002 D5                         PUSH D          ;SAVE DE
 0003 E5                         PUSH H          ;SAVE HL

                       ;     LOAD FORCE COMMAND AND
                       ;     SETUP FOR GAP MEASUREMENT
 0004 2A0000     E               LHLD ADCIN      ;LOAD FORCE(N)
 0007 3E06                       MVI A,6H        ;LOAD GAP CHANNEL
 0009 320000     E               STA MUXADR      ;SELECT GAP CHANNEL
 000C 320000     E               STA CONV        ;START GAP CONVERSION

                       ;     GENERATE DELTA FORCE COMMAND
                       ;     AND TABLE ADDRESS OFFSET
 000F 0600                       MVI B,00H       ;CLEAR B
 0011 7C                         MOV A,H         ;HIGH FORCE DATA > A
 0012 E60F                       ANI 0FH         ;MASK OUT HIGH 4 BITS
 0014 67                         MOV H,A         ;STORE DB11 THRU DB8 > H
 0015 7D                         MOV A,L         ;LOW FORCE DATA > A
 0016 E61F                       ANI 1FH         ;MASK OUT HIGH 3 BITS
 0018 4F                         MOV C,A         ;DELTA FORCE > C
 0019 AD                         XRA L           ;MOVE LOW DATA > A (LOW 5 BITS M
                         ASKED)
 001A 84                         ADD H           ;COMBINE DB11 THRU DB8 WITH DB7
                         THRU DB5
 001B 07                         RLC             ;
 001C 07                         RLC             ;REARRANGE BYTE
 001D 07                         RLC             ;   TO FORM
 001E 07                         RLC             ;    DB10,DB9,DB8,DB7,DB6,DB5,0,
                         0
 001F 17                         RAL             ;
 0020 6F                         MOV L,A         ;CONVERT TO
 0021 3F                         CMC             ; OFFSET
 0022 17                         RAL             ;  BINARY
 0023 E601                       ANI 01H         ;   FORM
 0025 67                         MOV H,A         ;

                       ;     COMPUTE ADDRESS OF SLOPE(N)
 0026 110000     E               LXI D,TABLE     ;LOAD TABLE START ADDRESS
 0029 23                         INX H           ;ADJUST "TABLE" ADDRESS VALUE
 002A 23                         INX H           ;
 002B 19                         DAD D           ;ADD TABLE START TO OFFSET


                       ;     LOAD SLOPE(N)
```

```
ISIS-II 8080/8085 MACRO ASSEMBLER, V3.0        MODULE   PAGE    2

 LOC  OBJ         LINE        SOURCE STATEMENT

0002C 5E           50         MOV E,M        ;LOAD LOW BYTE SLOPE(N)
002D 23            51         INX H          ;INCREMENT POINTER FOR HIGH BYTE
002E 56            52         MOV D,M        ;LOAD HIGH BYTE SLOPE(N)
002F 23            53         INX H          ;INCREMENT POINTER FOR CURRENT A
                    DDRESS
                   54
                              55 ;     MULTIPLY SLOPE(N) BY DELTA FORCE
                              56 ;     (SPECIAL 8BIT X 16BIT MULT. )
0030 E5            57         PUSH H         ;SAVE ADDRESS
0031 EB            58         XCHG           ;SWAP MULTIPLICAN(SLOPE) TO HL
0032 220000        59         SHLD TEMP      ;SAVE MULTIPLICAN IN TEMP
0035 21CE00        60         LXI H,BNUM     ;LOAD CYCLE COUNTER ADDRESS
0038 3609          61         MVI M,09H      ;LOAD CYCLE COUNTER
003A 110000        62         LXI D,0H       ;CLEAR TEMPORARY RESULT
003D 79            63 LOOP:   MOV A,C        ;
003E 1F            64         RAR            ;ROTATE MULTIPLIER(DELTAF)
003F 4F            65         MOV C,A        ;
0040 35            66         DCR M          ;DECREMENT CYCLE COUNTER
0041 CA5800        67         JZ FINI        ;TEST FOR MULTIPLY COMPLETE
0044 D24F00        68         JNC SKIP       ;JUMP IF MULTIPLIER BIT 0
0047 2A0000        69         LHLD TEMP      ;GET MULTIPLICAN
004A 19            70         DAD D          ; AND ADD
004B EB            71         XCHG           ;SAVE PARTIAL PRODUCT
004C 21CE00        72         LXI H,BNUM     ;RELOAD BNUM ADDRESS
004F 7A            73 SKIP:   MOV A,D        ;
0050 1F            74         RAR            ;ROTATE
0051 57            75         MOV D,A        ;
0052 7B            76         MOV A,E        ;  TEMPORARY
0053 1F            77         RAR            ;
0054 5F            78         MOV E,A        ;   RESULT
0055 C33D00        79         JMP LOOP       ;LOOP
0058 E1            80 FINI:   POP H          ;RESTORE ADDRESS
                   81
                              82 ;     COMPUTE CURRENT COMMAND FROM
                              83 ;     BASEI(N) AND DELTA I
0059 4E            84         MOV C,M        ;LOAD LOW BYTE BASEI(N)
005A 23            85         INX H          ;
005B 46            86         MOV B,M        ;LOAD HIGH BYTE BASEI(N)
005C EB            87         XCHG           ;SWAP DELTAI TO HL
005D 09            88         DAD B          ;ADD BASEI(N) TO DELTAI
005E 44            89         MOV B,H        ;MOVE CURRENT COMMAND
005F 4D            90         MOV C,L        ;  TO BC
                   91
                              92 ;     CORRECT FOR BEARING GAP VARIATION
0060 3A0000   E    93         LDA STATUS     ;CLEAR RTI1200 EOC FLAG
0063 2A0000   E    94         LHLD ADCIN     ;LOAD GAP
0066 29            95         DAD H          ,DOUBLE GAP
0067 3E05          96         MVI A,5H       ;LOAD FORCE COMMAND CHANNEL
0069 320000   E    97         STA MUXADR     ;SELECT FORCE COMMAND CHANNEL
006C 111621        98         LXI D,GAP      ;LOAD NOMINAL GAP VALUE
006F B0            99         ORA B          ;TEST FOR SIGN OF CURRENT COMMAN
                    D
0070 F5           100         PUSH PSW              ;SAVE SIGN FLAG
0071 F27F00       101         JP UPPER       ;JUMP TO UPPER BEARING COIL
                  102
```

APPENDIX A

```
ISIS-II 8080/8085 MACRO ASSEMBLER, V3.0        MODULE   PAGE    3

 LOC  OBJ         LINE        SOURCE STATEMENT

                              103 ;     COMPUTE LOWER BEARING COIL GAP
 0074 19           104 LOWER:  DAD D          ;ADD GAP(T) TO NOMINAL GAP
 0075 0B           105         DCX B          ;
 0076 79           106         MOV A,C        ;TAKE
 0077 2F           107         CMA            ;  ABSOLUTE
 0078 4F           108         MOV C,A        ;
 0079 78           109         MOV A,B        ;   VALUE
 007A 2F           110         CMA            ;    OF
 007B 47           111         MOV B,A        ;   CURRENT
 007C C38500       112         JMP FINAL      ;JUMP TO FINAL CURRENT COMMAND C
                       OMPUTATION
                   113
                   114
                              115 ;     COMPUTE UPPER BEARING COIL GAP
 007F 7B           116 UPPER:  MOV A,E        ;SUBTRACT
 0080 95           117         SUB L          ;  GAP(T)
 0081 6F           118         MOV L,A        ;  FROM
 0082 7A           119         MOV A,D        ;  NOMINAL
 0083 9C           120         SBB H          ;  GAP AT
 0084 67           121         MOV H,A        ;  CENTER
                   122
                              123 ;     MULTIPLY GAP BY CURRENT COMMAND
 0085 220000       124 FINAL:  SHLD TEMP      ;STORE MULTIPLICAN IN TEMP
 0088 21CE00       125         LXI H,BNUM     ;STORE
 008B 3611         126         MVI M,11H;     ; BIT COUNT
 008D 110000       127         LXI D,00H      ;INITALIZE RESULT
 0090 78           128 LOOP1:  MOV A,B        ;ROTATE
 0091 1F           129         RAR            ;
 0092 47           130         MOV B,A        ;MULTIPLIER
 0093 79           131         MOV A,C        ;
 0094 1F           132         RAR            ;RIGHT
 0095 4F           133         MOV C,A        ;
 0096 35           134         DCR M          ;DECREMENT BIT COUNT
 0097 CAAE00       135         JZ FINI1       ;DONE? THEN OUTPUT
 009A D2A500       136         JNC SKIP1      ;JUMP IF NO CARRY FROM ROTATE
 009D 2A0000       137         LHLD TEMP      ; OTHERWISE
 00A0 19           138         DAD D          ; ADD MULTIPLICAN
 00A1 EB           139         XCHG           ;SAVE RESULT
 00A2 21CE00       140         LXI H,BNUM     ;RESTORE BIT COUNT POINTER
 00A5 7A           141 SKIP1:  MOV A,D        ;ROTATE
 00A6 1F           142         RAR            ; TEMP
 00A7 57           143         MOV D,A        ; RESULT
 00A8 7B           144         MOV A,E        ; RIGHT
 00A9 1F           145         RAR            ;
 00AA 5F           146         MOV E,A        ;
 00AB C39000       147         JMP LOOP1      ;REPEAT LOOP
                   148
                              149 ;     CHECK FOR OVERCURRENT VALUE
                              150 ;     AND OUTPUT RESULT TO CORRECT PORT
 00AE EB           151 FINI1:  XCHG           ;SWAP CURRENT TO HL
 00AF 7C           152         MOV A,H        ;LOAD CURRENT
 00B0 FE08         153         CPI 08H        ;TEST CURRENT VALUE
 00B2 DAB800       154         JC OK          ;JUMP IF OK
 00B5 21FF07       155         LXI H,07FFH    ;CURRENT=MAXIMUM CURRENT
 00B8 F1           156 OK:     POP PSW        ;RESTORE CURRENT SIGN
```

APPENDIX A

```
ISIS-II 8080/8085 MACRO ASSEMBLER, V3.0        MODULE   PAGE    4

 LOC  OBJ         LINE        SOURCE STATEMENT

 00B9 F2C200       157         JP OUTPOS       ;JUMP IF POSITIVE
 00BC 220000   E   158         SHLD CURNEG     ;OUTPUT RESULT TO LOWER BEARING
                       D/A
 00BF C3C500       159         JMP RETURN      ;RETURN
 00C2 220000   E   160 OUTPOS: SHLD CURPOS     ;OUTPUT RESULT TO UPPER BEARING
                       D/A
 00C5 E1           161 RETURN: POP H           ;RESTORE HL
 00C6 D1           162         POP D           ;RESTORE DE
 00C7 C1           163         POP B           ;RESTORE BC
 00C8 3E20         164         MVI A,20H       ;LOAD END OF INTERUPT
 00CA D3DA         165         OUT 0DAH        ;OUTPUT EOIC
 00CC F1           166         POP PSW         ;POP A & PSW
 00CD C9           167         RET             ;RETURN
                   168
                   169
                   170
                   171
 00CE 0000         172 BNUM:   DW      0
 00D0 0000         173 TEMP:   DW      0
 2116              174 GAP     EQU     8470
                   175         END

PUBLIC SYMBOLS


EXTERNAL SYMBOLS
ADCIN  E 0000    CONV   E 0000    CURNEG E 0000    CURPOS E 0000
MUXADR E 0000    STATUS E 0000    TABLE  E 0000

USER SYMBOLS
ADCIN  E 0000    BNUM   A 00CE    CONV   E 0000    CURNEG E 0000
CURPOS E 0000    FINAL  A 0085    FINI   A 0058    FINI1  A 00AE
GAP    A 2116    INTRU  A 0000    LOOP   A 003D    LOOP1  A 0090
LOWER  A 0074    MUXADR E 0000    OK     A 00B8    OUTPOS A 00C2
RETURN A 00C5    SKIP   A 004F    SKIP1  A 00A5    STATUS E 0000
TABLE  E 0000    TEMP   A 00D0    UPPER  A 007F

ASSEMBLY COMPLETE,   NO ERRORS
```

APPENDIX B

MAGNETIC BEARING CURRENT DRIVER

A schematic diagram of the current driver used in the magnetic bearing test system is shown in figure B1. The driver is capable of supplying up to



Figure B1.- Magnetic bearing current driver.

5 A to each bearing coil and has a gain of 1 A/V. Drive is provided by the LH 0002 current amplifier at low current levels (<100 mA) and by the complementary Darlington configuration at higher levels. The driver provides flat response over a bandwidth greater than 200 Hz (fig. B2). The frequency response is limited by the ability of the power supply voltage to overcome the back emf of the bearing coil, and the effect is observed as an apparent slew rate limitation. The 200-Hz bandwidth is sufficient to evaluate the table look-up algorithm.

APPENDIX B



(a) Magnitude.



(b) Phase.

Figure B2.- Frequency response (magnitude and phase) of magnetic bearing current driver.